積分の数値計算アラカルト（入門ノート）

SHIMURA Masato
jcd02773@nifty.ne.jp

2010/12/04

目次

概要

ニュートン · コーツ法を中心に数値積分の手法と、乱数やディジタルデータの数値積分の諸々の手法をレビューした。

準備

J のパッケージ

```
require '~system/packages/math/integrat.ijs'
```

本稿は区間を指定した積分を前提としている。数学語「∞、連続、有界、上限、下限など」の論議は専門書に依られたい。

1　積分と数値計算

数値積分には Newton-Cotes と Gauss の系譜がある。

Sir Isaac Newton(1642 – 1727)

Roger Cotes(1682 – 1716) ゲンブリッジの天文と数学の教授。Newton のプリンキピア（第 2 版）

の編集にも携わった。

Z) いきなりすごい人が出てきましたが

A) 数値積分の原点

1.1　Newton-Cotes 法の系譜

森　p169

「$\int_0^1 x^2 = \dfrac{1}{3}$ となると言うことを今までに検討した人は稀でしょうから」

Z) やってみましょう

ニュートン · コーツの台形公式

$$I = \int_a^b f(x)dx = \frac{h}{2}\left(f(x_0) + 2\sum_{j=1}^{n-1} f(x_j) + f(x_n)\right) - \frac{(b-a)h^2}{12} f''\xi \tag{1}$$

$$I(f) = \sum_{j=1}^{n} \int_{x_{j-1}}^{x_j} f(x)dx$$

$$= \sum_{j=1}^{n} \frac{x_j - x_{j-1}}{2}[f(x_{j-1}) + f(x_j)] - \sum_{j=1}^{n} \frac{(x_j - x_{j-1})^3}{12} f''(\xi_j)$$

$$= \frac{h}{2}\left(f(x_0) + 2\sum_{j=1}^{n-1} f(x_j) + f(x_n)\right) - \frac{h^3}{12}\sum_{j=1}^{n} f''(\xi_j)$$

$-\dfrac{h^3}{12}\sum_{j=1}^{n} f''(\xi_j)$ は誤差項である。誤差項は式の精度を示すので誤差項を付けた式も良く用いられる。

Z) この台形公式は一次式（直線）で近似したものと言われるが、単に（引数の）導関数の下の面積を計算したものではないのか。

A) ライプニッツの積分の目的はは曲線の下の求積ではなかったかな。（計算には向いていないが）

積分の数式処理は $\int_a^b$ と dx で囲った部分の処理であるが、数値計算では引数の数式の数式処理などには目もくれず、（導）関数の下部の求積をいかに効率的に正確に行うかに専念する。単純な関数の下の求積 Script も作ってみよう。

A) 数値計算では数式はポイントの値しか取れないのでポイント間を補完した方が精度が良くなる。台形公式は些かシンプルな直線近似だが今の高性能 PC で分割を小さくすれば精度は言われるほど程悪くない。

f_0 と f_n は端なので 2 倍しない。

A) テーラー展開から台形公式やシンプソン法を導くこともできる。（appendix 参照）

線形数学からもアプローチできるが手計算は困難を極める。兎に角曲線の下の面積を求める方法はニュートン · コーツ法のイングランド系とガウス法が 2 大潮流だ。

Z)Newton-Cotes も多項式の次数を上げげたものが沢山提案されており、300 年経ても現役だ。この系のクラッシクはシンプソン法で 2 次多項式で近似する

Simpson method

$$\int_{x_1}^{x_2} f(x)dx = \int_{x_1 1}^{x_1+2h} P_3(x)dx = \frac{1}{3}(f_1 + 4f_2 + f_3) - \frac{1}{90}f^{(4)}(\xi) \tag{2}$$

$\frac{1}{90}f^{(4)}(\xi)$ は誤差項

A) 数学では Δx とか $lim_{x\to 0}$ などと思考を巡らすが、数値計算では極端な細分化は不安定となる。最近の数値計算言語は分数をサポートし、桁数制限も無くなっており、割り算の誤差が大幅に改善されているので細分化に拘ることはない。

■Example 1

手計算　.

$$\int_3^4 x^3 dx = \left.\frac{x^4}{4}\right|_3^4 = 43.75$$

Z) 積分した後は 3 ,4 と積分区間の両端を上限、下限として大雑把に分割しているが解は Exact である

```
 -/  (4 3 ^4) % 4
43.75
```

台型則

```
  ^&3 daikei 3 4 10
43.7675


   ^&3 daikei 3 4 100
43.7502
```

Script　.

```
   daikei=: 1 : 0
NB. Newton_Cotes Method
```

```
'A0 B0 N0'=. y
xn =: A0, A0+ (B0-A0)* (>: i. N0) % N0
dh=: ((B0-A0)% N0) * 1r2          NB. delta
dh*  +/({. tmp),(+: +/ }. }: tmp),{: tmp=: u xn
)
```

simpson0（自作） 移動平均を忠実にプログラムした simpson0 を用いる。分割を細かくしないと精度がでない

```
      ^&3 simpson0 3 4 100
43.2955


   ^&3 simpson0 3 4 10000
43.7455
```

Script .

```
simpson0=: 1 : 0
NB. moving average type
NB. usage: ^&3 simpson0 3 4 100
dh=: 1r6 * (-/1 0 { y) %  {:y          NB. delta
dh * +/ ; +/ (L:0) 1 4 1 * (L:0) u (L:0) 3<\ xn=. steps y
)
```

Simpson 法 (J のパッケージ)

(3, 4) を 6 分割でも EXACT

```
   ^&3 simpson 3 4 10
43.75
```

Script J の package を真似たもの

```
simpson1=: 1 : 0
NB. weight is 1 4(odd) 2(even) 1
dh=: (-/ 1 0 { y) %  {: y   NB. delta
tmp=: dh * u  xn=. steps y
+/ 1r3 * tmp * 1,((<: {: y)$ 4 2),1
)
```

2　種々の積分と数値計算

2.1　数値積分と差分

■Example　A) 次のような 6 本のローソク図形を頭の体操で計算してみよう。(厳密な階段積分ではない)

$$\begin{bmatrix}5 & 6 & 7 & 8 & 9 & 10\end{bmatrix}$$

$\Delta x = 1$

$f = 3 + x$

```
 3 1&p. i.10
```



```
(>:i.8),. 0 5 6 7 8 9 10 0
1  0
2  5
3  6
4  7
5  8
6  9
7 10
8  0
```

前進差分、後退差分　.

```
   a=.5 6 7 8 9 10
   2<\a
+---+---+---+---+----+
|5 6|6 7|7 8|8 9|9 10|
+---+---+---+---+----+
;(+/%#) (L:0) 2<\a NB. 平均
5.5 6.5 7.5 8.5 9.5

 +/
37.5

sabun0=:3 : ';(+/%#) L:0  |.(L:0) 2<\ y'
```

中心化差分　3 項づつ取って中央を落とす

```
 0 2 {(L:0)  3<\a
```

```
+---+---+---+----+
|5 7|6 8|7 9|8 10|
+---+---+---+----+
;(+/%#) (L:0) 0 2 {(L:0)  3<\a
6 7 8 9  NB. 平均
```

$$\frac{5+2(6\ 7\ 8\ 9)+10}{2}=\frac{75}{2}=37.5$$

Z) 台型法と同じではありませんか

ア・ラ・メゾン .

A) 折角計算した元データも使うと

$$\sum\begin{bmatrix}5 & 6 & 7 & 8 & 9 & 10\end{bmatrix}=45$$

$$\frac{45+6\ 7\ 8\ 9}{2}=\frac{75}{2}=37.5$$

台型法の手計算　中心差分と同等の手法

```
 ({@> 1 2 1) *(L:0)  5;6 7 8 9;10
+-+-----------+--+
|5|12 14 16 18|10|  NB. divide side and center
+-+-----------+--+
;+/ L:0 ({@> 1 2 1) *(L:0)  5;6 7 8 9;10
+-+--+--+
|5|60|10| NB. sum on each Box
+-+--+--+
2%~+/ ;+/ L:0 ({@> 1 2 1) *(L:0)  5;6 7 8 9;10
37.5   NB. 75%2
```

台形法の数値計算　　`f=.3 1&p.   NB. f=3+x`

```
   f daikei 2 7 5
37.5
```

数学の積分

$$\int_2^7 (3+x)dx = 3x+\left.\frac{x^2}{2}\right|_2^7 = 45.5-8=37.5$$

2.2 畳み込み積分

Z) 数値計算では、寒天で作った柔らかい電車 2 両を衝突させるイメージ

- 手計算の Example
  (出典:geocities.jp/maeda_hashimoto/math/sig_ch02pr00p01.htm)

$$g(t) = \begin{bmatrix} 4 & 2 & 1 & 0.7 \end{bmatrix}$$
$$x(t) = \begin{bmatrix} 1 & 2 & 3 & 4 \end{bmatrix}$$

$$y(t) = g(t) * x(t) \quad \text{離散信号の畳み込み和}$$

			0.7	1	2	4
		×	4	3	2	1
			0.7	1	2	4
		1.4	2	4	8	
	2.1	3	6	12		
2.8	4	8	16			
2.8	6.1	12.4	24.7	17	10	4

A) 畳み込みの計算に J のイディオム *ppr* が利用できる

- ppr

```
ppr=: +//.@(*/) NB. multiplication of 2 polinomials
```

$p(x) = 1 + 2x^2$
$q(x) = 34x + x^2$
$p(x)q(x) = -3 - 4x - 5x^2 - 8x^3 + 2x^4$

```
      1 0 2 ppr _3 _4 1
_3 _4 _5 _8 2
```

- ベクトルのテーブル演算

```
      0.7 1 2 4  (*/) table 4 3 2 1
+---+---------------+
|*/ |  4   3   2   1|
+---+---------------+
|0.7|2.8 2.1 1.4 0.7|
|  1|  4   3   2   1|
|  2|  8   6   4   2|
|  4| 16  12   8   4|
+---+---------------+
```

- Oblique(`/.`)
  Z) 薩摩示現流右袈裟懸けで右上から斜めに切断

```
     0.7 1 2 4 </.@(*/) 4 3 2 1
+---+-----+-------+---------+-----+---+-+
|2.8|2.1 4|1.4 3 8|0.7 2 6 16|1 4 12|2 8|4|
+---+-----+-------+---------+-----+---+-+
```

- ボックス内の和

```
   0.7 1 2 4 ppr 4 3 2 1
2.8 6.1 12.4 24.7 17 10 4
```

2.3　部分積分

部分積分の公式

$$\int f'(x)g(x) = f(x)g(x) - \int f(x)g'(x)dx$$

*1

■Example　.

$$\int e^x x dx$$

$$\int e^x x dx = \int (e^x)' x dx = e^x x - \int e^x (x)' dx = e^x x - \int e^x dx = e^x x - e^x + C$$

$$\int x^2 e^x dx$$

$$\int x^2 e^x dx = \int x^2 (e^x)' dx = x^2 e^x - \int 2x e^x dx = x^2 e^x - 2\int x (e^x)' dx$$

$$= x^2 e^x - 2\left(x e^x - \int e^x dx \right) = x^2 e^x - 2x e^x + 2e^x + C$$

---

*1 定積分の場合

$$\int f'(x)g(x) = [f(x)g(x)]_a^b - \int f(x)g'(x)dx$$

積の微分の公式

$$(f(x) \cdot g(x))' = f'(x) \cdot g(x) + f(x) \cdot g'(x)$$

積分記号を付けて整理する。
(微分記号の付け方が 2 通りあり、$f(x), g(x)$ のどちらかに（交互に）付く）

$$\int f'(x) \cdot g(x)dx = f(x) \cdot g(x) - \int f(x) \cdot g'(x)$$

Z) 大変紛らわしい

- **Example**（物理のかぎしっぽから）

$$\int x\ sinx\ dx$$

- 一度 2 に分けて微分の練習を行い

$$x' = 1$$

$$sinx' = -cos\ x$$

- 片方（後方）を微分型にして馬揃えをおこなう

$$\int x\ sinx\ dx = \int x(-cosx)'dx$$

- 部分積分へ
  - (1) 外へ出して (2) 微分を $f(x) \leftarrow g(x)$ へ

$$= x(-cosx) - \int (x)'(-cosx)dx$$

  - 微分（と外だし）

$$= -x\ cosx + \int cosx\ dx$$

  - 積分して整理

$$= -xcosx + sinx + C$$

A) スティエルチェス積分などで類似が出てくるから馴染んでおこう

2.4　スティルテェス積分

2.4.1　関数 · 関数の場合

$$\int_a^b f(x)d\varphi(x) = \int_a^b f(x)\varphi'(x)dx$$

$$\int_a^b f(x)d\varphi(x) = [f(x)g(x)]_a^b - \int_a^b g(x)df(x)$$

■Example　.

$f(t) = t^3$

$\alpha(t) = t^2$

区間　 $[1, 5]$

$$\int_1^5 t^3 d(t^2)$$

φ の項を 1 回微分した後合算してから積分する。[*2]

$$= \int_1^5 t^3(t^2)dt = \int_1^5 2t^4 dt = \left.\frac{2t^5}{5}\right|_1^5 = \frac{6248}{5}$$

■Example　.

$$\int_0^1 x^2 d(x^3)dx = \int_0^1 x^2(3x^2)dx = \int_0^1 3x^4 dx = \left.\frac{3}{5}x^5\right|_0^1 = \frac{3}{5}$$

■Example　分離する場合

$$\int_0^3 [x]d(x^2) = \int_0^1 (0)(2x)dx + \int_1^2 (1)(2x)dx + \int_2^3 (2)(2x)dx = x^2\,|_1^2 + 2x^2\,|_2^3 = 3 + 10 = 13$$

Z）スチルテェス積分のイメージが掴みにくい

A) 座標の X 軸は普通は平坦で、数値積分は差分を考えるので等間隔だが、鳥瞰図の様に上から眺めたらどうだろう。X 軸が崖で X 軸を崖に沿わせて持ち上げ、再度 X 軸に投影すると X 軸の $x_0, x_1 \cdots$ の位置が変更される。これが $g(x)$

Z)　鳥から見た積分！

[*2] 3step で畳み込み

2.4.2　関数 · ディジタルデータの階段積分

■Example　（小林　p83）

次のリーマン · スティルテェス積分を求める

$$\int_0^6 f(x)d\varphi(x)$$

$$f(x) = x^2 NB.\text{関数} = \text{鳥で自由に飛び回る}$$

$$\varphi(x)\ \text{右図のディジタルデータ}$$



$$\int_0^6 f(x)d\varphi(x) = 2f(1) + 3f(2) + 1(5) + 1f(6)$$

1 で 2、2 で 3、5 で 1 そして 6 で 1 ジャンプした階段積分

A) どのように入力フォームを決めるか

Z) 次の縦型でどうでしょう

$$LS\ =.\ \begin{array}{cc} x & \varphi x \\ 1 & 2 \\ 2 & 3 \\ 5 & 1 \\ 6 & 1 \end{array}$$

$f(x) = x^2$ と階段の数値 (φ) をかける

```
  (*: 1 2 5 6)*2 3 1 1   NB. *: (squre)
2 12 25 36
```

```
 +/ (*: 1 2 5 6)*2 3 1 1 NB. sum
75


     ^&2 ls_integral0 LS0
75
```

```
NB. Kobayashi p83
ls_integral0=: 1 : ' +/ (u {."1 y) * {:"1 y'
NB. fx is ^&2
NB. varphi is LS0


LS0=: 1 2 5 6,. 2 3 1 1
```

Z) 鳥の軌跡 ($f(x)$) は普通のリーマン積分のようだが、階段の ($\varphi(x)$) のジャンプの意味が分かりづらい

A) ジャンプは $f(x)$ の差分である。これを縦に取ると面積にならない。この差分を手旗信号のように 90^o 廻して横に上げたらどうだろう。ジャンプしないところは 0 になる。細分化して極限をもとめると積分になる。

Z) マイナスの差分はどうするか面積はある。

A) マイナスする

3　ディジタルデータと積分

3.1　ブラウン運動と伊藤積分

■Example　ブラウン運動 $B(t,\varpi)$ のサンプルパス

(出典：小林 P83)



伊藤積分の近似値を求める

$I0 = .$

$B(0,\varpi)$	0
$B(1,\varpi)$	0.55
$B(2,\varpi)$	0.84
$B(3,\varpi)$	_0.8
$B(4,\varpi)$	1.03
$B(5,\varpi)$	_0.04
$B(6,\varpi)$	1.41
$B(7,\varpi)$	_0.56
$B(8,\varpi)$	0.66
$B(9,\varpi)$	1.78
$B(10,\varpi)$	_0.45

$$\int_0^{10} B(t,\varpi)dB(t,\varpi)$$

Z) $B(t,\varpi)$ が 2 回現れる

- 差分用のテーブル (一つずらしたものを右に append)

```
]a=.(tmp,.-/"1 tmp=. (}. I0),.}:I0),. */("1) 0 2{"1 a
 }.IO  }:IO  差分  }.I0*差分
 0.55     0  0.55 0.3025
 0.84  0.55  0.29 0.2436
 _0.8  0.84 _1.64  1.312
 1.03  _0.8  1.83 1.8849
_0.04  1.03 _1.07 0.0428
 1.41 _0.04  1.45 2.0445
_0.56  1.41 _1.97 1.1032
 0.66 _0.56  1.22 0.8052
 1.78  0.66  1.12 1.9936
_0.45  1.78 _2.23 1.0035
```

- $\sum$

```
   +/ a
4.42 4.87 _0.45 10.7358
```

- ito_integral

```
   ito_integral_2 I0
10.7358
```

- Script

```
ito_integral_2=: 3 : '+/ (}.y)* -/"1 tmp=: }: (}.y,0),.y'
```

3.2 ウインナー過程

ランダムウオークが一歩づつの ± の累積であるのに対してブラウン運動は時間を連続にして振幅を無限にする。実際、ランダムウオークは　0, 1 の乱数を用いるのに対してブラウン運動は平均 =0, 分散=1 の正規乱数を用いる。

Z）どれくらい異なるか。

A) ランダムウオークはプロペラ機の上昇下降。ブラウン運動はジェット機。




```
winner0=: 3 : 0
NB. Winner integral
NB. Usage:  plot winner0 0 1 256 //256=2^8 inverse(%)=2^_8
dh=: %: (-/1 0{ y) % {: y  NB. Delta_h --> use sqrt(h)
+/\ dh * normalrand {: y   NB. normalrand mean=0 sd=1
)
```

幅	$\sqrt{dh}$	`%: (b-a) % N`
正規乱数	$normalrandN$	
縦長の板	$dh * normanrandN$	
積分（累積和）	`+/\`	

3.3 ディジタルデータを Newton·Cotes 法で

リーマン積分 短冊和

$$\int_a^b f(x)dx = \sum_{i=0}^{N-1} f(x_i)\Delta x$$

台型則（ディジタル版） 2.1 参照

x（左引数) は Δ

```
 1  daikei_digital a=:5 6 7 8 9 10
37.5
```

1.1 参照 $\int_3^4 x^3$

```
0.01 daikei_digital  ^&3 steps 3 4 100
43.7502
```

Script Newton-Cotes 法での両端処理は不要なので 2 倍もしない

```
   2<\a
```

```
+---+---+---+---+----+
|5 6|6 7|7 8|8 9|9 10|
+---+---+---+---+----+
```

```
daikei_digital=: 4 : 0
NB. Newton_Cotes Method
NB. x is 1 or dh
tmp=: ;(+/%#) (L:0)  2<\y  NB. mean of both side
dh=. x
+/ dh*tmp
)
```

シンプソン法（ディジタル版） .

```
0.01 simpson_digital  ^&3 steps 3 4 100
43.7512
```

Script .

```
simpson_digital=: 4 : 0
tmp=. (({.y), {:y) , 2* ;(+/ % #)(L:0) 0 2 { (L:0) 3<\y
-: +/ x * tmp
)
```

4 J と数値積分

4.1 原始関数

d.n d._n 微分と逆関数としての積分

- **Examples**

$$f = x^3$$

x0=: 1 2 3 4 5

```
   cube=. 0 0 0 1&p.  NB. polinomial x^3
   cube x0=. 1 2 3 4 5
1 8 27 64 125
```

- 差分 (D:n) n は階数

```
   0.01 cube D:1 x0
3.0301 12.0601 27.0901 48.1201 75.1501
```

- 微分

```
   cube d.1 x0
3 12 27 48 75
```

- 積分（J は微分の逆関数として定義）

```
   cube d._1 x0
0.25 4 20.25 64 156.25
```

多項式の微分と積分　p..

J はもう一つ多項式の微積分の原始関数を持っている

$$f = 1 + 3x + 3x^2 + x^3$$

- 微分 (単項)

$$f' = 3 + 6x + 3x^2$$

```
   p.. 1 3 3 1
3 6 3
```

- 積分

  左引数は積分定数 C(係数のみの計算)

$$3 + x + \frac{3x^2}{2} + \frac{3x^3}{3} + \frac{x^4}{4}$$

```
   3 p.. 1 3 3 1
3 1 1.5 1 0.25
```

4.2 Packages

```
system/packages/math/integrat.ijs
```

```
(1) integrate     Aitken extrapolation on Gauss integrals
(2) simpson       Simpson's method
(3) adapt         Adaptive Quadrature using Simpson's method
```

integrate　エイトケン外挿付きガウス法
simpson　シンプソン法 ($\frac{1}{3}$)
adapt　上のシンプソン法を用いた適応的積分（ロンバーク法）

付録 A　テーラー展開から台形公式を求める

$$I = \int_a^b f(x)dx \approx \frac{h}{2}\left(f_0 + 2\sum_{i=1}^{n-1} f_i + f_n\right) \tag{3}$$

佐藤 · 中村

$$f_{i+1} = f_i + hf_i' + \frac{h^2}{2!}f_i'' + \frac{h^3}{3!}f_i''' + \frac{h^4}{4!}f_i^4 + \cdots$$

$$f_i' = \frac{f_{i+1} - f_i}{h} - \frac{h}{2}f_i'' - \frac{h^2}{6}f_i''' - \frac{h^4}{24}f_i^4 - \cdots \tag{4}$$

$$I = \int_{x_i}^{x_{i+1}} f(x)dx = \int_0^h f(x_i + z)dz = \int_0^h \left(f_i + zf_i' + \frac{z^2}{2!}f_i'' + \frac{z^3}{3!}f_i''' + \frac{z^4}{4!}f_i^4 + \cdots\right)dz$$

$$= hf_i + \frac{h^2}{2}f_i' + \frac{h^3}{6}f_i'' + \frac{h^4}{24}f_i''' + \frac{h^5}{120}f_i^4 + \cdots$$

$$I_i = h\frac{f_{i+1} + f_i}{2} - \frac{h^3}{12}f_i'' - \frac{h^5}{120}f_i^4 - \cdots$$

References

小林道正「ブラック · ショルズと確率微分方程式」朝倉書店　2003
佐藤次男 · 中村理一郎「よくわかる数値計算」日刊工業新聞社　2001
森真「なっとくする数理ファイナンス」講談社 2001